

OBSZORY

4/2003(32)

MATEMATIKY
FYZIKY a
INFORMATIKY

OBZORY MATEMATIKY, FYZIKY A INFORMATIKY 4/2003(32)

Časopis pre teóriu a praktické otázky vyučovania matematiky,
fyziky a informatiky na základných a stredných školách

Vedeckí redaktori:

matematickej časti: Beloslav R i e č a n

fyzikálnej časti: Daniel K l u v a n e c

Výkonní redaktori:

matematickej časti: Peter M a l i č k ý

fyzikálnej časti: Aba T e l e k i

Redakčná rada:

Matematická a informatická časť:

Hynek Bachratý, Viera Blahová, Pavol Černek, Jozef Doboš, Martin Gavallec, Tomáš Hecht, Karel Horák, Vladimír Jodas, Mária Kmeťová, Viera Kyselicová, Marian Macko, Peter Maličský, Božena Mihalíková, Anna Michalcová, Gustáv Nagy, Zdeněk Půlpán, Mária Sadloňová, Bohuš Sivák, Robert Szelepcsényi, Ladislav Topolský, János Tóth, Jan Vinař, Michal Winczer, Viktor Witkovský

Fyzikálna časť:

Mária Barbierová, Michal Blaško, Anna Jankovychová, Václav Havel, Arpád Kecskés, Dalibor Krupa, Václav Koubek, Stanislav Ondrejka, Anna Pribišová, Mária Rakovská, Juraj Šebesta, Eva Tomanová, Ivo Volf

Adresa redakcie:

Matematická časť:

Katedra matematiky FPV UMB, Tajovského 40, 974 01 Banská Bystrica
(e-mail: malicky@fpv.umb.sk)

Fyzikálna časť:

Katedra fyziky, FPV UKF, Tr. A. Hlinku 1, 949 74 Nitra
(e-mail: ateleki@ukf.sk)

Objednávky a predplatné vybavuje:

Redakcia OMFÍ, FPV UKF, Tr. A. Hlinku 1, 949 74 Nitra

gnuplot – grafický software (aj) pre matematikov

Jozef Doboš

Abstract: gnuplot is a free command-line driven interactive function and data plotting program. It can be run under Windows, Macintosh OS, VMS, Linux, and many others. gnuplot can send the result to a wide range of graphic devices (terminals, printers, or plotters). The aim of the paper is to present this interesting program.

Úvod

Tento program možno získať na adrese


<http://www.gnuplot.info/>

K dispozícii je interaktívny help, ako aj 120 stranový manuál.

Ihneď po spustení programu môžeme kresliť. Začneme grafom exponenciálnej funkcie $y = e^x$. Do príkazového riadku napíšeme

```
plot exp(x)
```

a stlačíme . Výsledok vidíme na obrázku 1.




Obrázok 1

gnuplot kreslí obrázok rozmerov 5 in × 3 in.

Podme si tento graf vylepšovať. Najskôr nakreslíme osi a zvolíme vhodnejšie intervaly pre premenné x a y . Graf je na obrázku 2.


```
set zeroaxis  
plot [-1:5] [-1:5] exp(x)
```



Obrázok 2

Štvorcový formát dosiahneme príkazom (pozri obrázok 3)

```
set size square
```




Obrázok 3


Teraz odstránime rôzne zbytočnosti pomocou príkazov

```
set nokey
set noborder
set noxtics
set noytics
```

Výsledok vidíme na obrázku 4.



Obrázok 4



Obrázok 5

Osi môžeme narysovať ako orientované úsečky pomocou príkazov

```
set arrow 1 from -1,0 to 5,0
set arrow 2 from 0,-1 to 0,5
```

Pozri obrázok 5.

Často sa opakujúce nastavenia si môžeme uložiť do súboru

`gnuplot.ini`.

Skúste si vytvoriť súbor `gnuplot.ini`, ktorý obsahuje nasledujúce riadky:


```
set nokey
set noborder
set noxtics
set noytics
set noztics
```

Skôr ako prejdeme k ďalšej ukážke, vyskúšajte si užitočný príkaz `reset`.

V nasledujúcej úlohe si nakreslíme graf funkcie tangens. Použijeme príkaz

```
plot tan(x)
```

Ako vidíme na obrázku 6, výsledok nie je najlepší.




Obrázok 6

Príkazmi

```
set sample 600
plot [-10:10] [-30:30] tan(x)
```

docielime prijateľnejší graf (pozri obrázok 7).




Obrázok 7

Vylepšíme ho pridaním osí a asymptot príkazmi

```
set arrow 1 from -10,0 to 10.7,0
set arrow 2 from 0,-30 to 0,33
set arrow 3 from pi/2,-30 to pi/2,30 nohead linetype 0
set arrow 4 from 3*pi/2,-30 to 3*pi/2,30 nohead linetype 0
set arrow 5 from 5*pi/2,-30 to 5*pi/2,30 nohead linetype 0
```

```
set arrow 6 from -pi/2,-30 to -pi/2,30 nohead linetype 0
set arrow 7 from -3*pi/2,-30 to -3*pi/2,30 nohead linetype 0
set arrow 8 from -5*pi/2,-30 to -5*pi/2,30 nohead linetype 0
```

Na obrázku 8 vidíme výsledok.



Obrázok 8

Svoju prácu si môžeme uložiť do súboru príkazom

```
save 'nazov.plt'
```

Existujúci súbor môžeme zavolať príkazom

```
load 'nazov.plt'
```

Otvoriť existujúci súbor na editovanie možno príkazom

```
! edit nazov.plt
```

Do súboru nazov.plt si môžeme písať aj svoje poznámky. Časť riadku, ktorá začína symbolom #, je komentár.

Pre zvýšenie pohodlia si v gnuplote môžeme vybrať iný typ písma. Po stlačení pravého tlačítka myši sa otvorí nasledujúca ponuka:

Copy to Clipboard
 Paste
 Choose Font ...
 System Colors
 Update wgnuplot.ini

V ponuke prejdeme myšou na

Choose Font ...


a vyberieme si vyhovujúci typ písma (vyskúšajte Courier, príp. Fixedsys).
Svoj výber uložíme pomocou

Update wgnuplot.ini

Grafy funkcií daných explicitne

Do jedného obrázku môžeme nakresliť aj niekoľko grafov funkcií. Ukážeme si to na príklade. (Pozri obrázok 9.)

```
set size square
set arrow 1 from -2.5, 0 to 2.5, 0 linetype 1
set arrow 2 from 0, -2.5 to 0, 2.5 linetype 1
set arrow 3 from pi/2, -2.5 to pi/2, 2.5 nohead linetype 0
set arrow 4 from -pi/2, -2.5 to -pi/2, 2.5 nohead linetype 0
plot [-2.5:2.5] [-2.5:2.5] exp(x), tan(x)
```



Obrázok 9


Ďalšia ukážka sa bude týkať funkcie

$$f(x) = \begin{cases} x^2 & -1 \leq x < 1 \\ 2-x & 1 \leq x \end{cases}$$


ktorej graf nakreslíme pomocou nasledujúcich príkazov:

```
set zeroaxis
f(x)= -1<=x && x<1 ? x**2 : 1<=x ? 2-x : 1/0
plot [-3:3] [-3:3] f(x)
```

Výraz 1/0 používame tam, kde funkcia nie je definovaná. Pozri obrázok 10.



Obrázok 10



Obrázok 11

Ten istý výsledok dosiahneme príkazmi:

```
set zeroaxis
f1(x)= x>=-1 ? x**2 : 1/0
f(x)= x<1 ? f1(x) : 2-x
plot [-3:3] [-3:3] f(x)
```

Nespojitá funkcie je vhodné rozdeliť na niekoľko častí. Napríklad graf funkcie

$$f(x) = \begin{cases} \frac{1}{2} \cdot (x + |x|) & x \leq 1 \\ 3 - x & x > 1 \end{cases}$$

môžeme nakresliť pomocou príkazov (pozri obrázok 11)

```
set arrow 1 from -1,0 to 2,0
set arrow 2 from 0,-1 to 0,2
set arrow 3 from 1, 0 to 1, 2 nohead linetype 0
set samples 500
f1(x)=x<=1 ? (x+abs(x))/2 : 1/0
f2(x)= x>1 ? 3-x : 1/0
plot [-1:2] [-1:2] f1(x), f2(x) lt 1
```

Ukážeme si, ako môžeme znázorniť polygonálne aproximácie Cantorovej funkcie (pozri obrázok 12). Vytvoríme si súbor pomocný cantor.plt obsahujúci príkazy

```
set samples 2000; set size square 0.5, 0.5
set zeroaxis lt 1; set xrange [0:1]; set yrange [0:1]
start(x)=x<=0?0:0<=x&&x<=1?x:1
f(a,x)=a==0?start(x):x<=1/3.?f(a-1,3*x)/2:f(a-1,3*x-2)/2+0.5
```

```

plot f(a,x)
a=a+1; pause -1
load 'cantor.plt'


```

a potom v gnuplote napíšeme príkazy

```

a=0
load 'cantor.plt'

```



Obrázok 12

Teraz si ukážeme, ako môžeme vyšrafovať elementárnu oblasť

$$a \leq x \leq b$$

$$f(x) \leq y \leq g(x)$$

Vyberieme si niečo jednoduché, napríklad

$$0 \leq x \leq 1$$

$$x^3 \leq y \leq x^2$$

Najskôr si vytvoríme pomocný súbor (napr. s názvom `re.plt`), ktorý obsahuje nasledujúce príkazy:

```

i = i+1
x(i)=a+i*(b-a)*n**(-1)
set arrow i from x(i), f(x(i)) to x(i), g(x(i)) nohead lt 0
if (i<n) reread


```

a potom v gnuplote napíšeme príkazy (pozri obrázok 13).

```

set nokey
set size square
a=0; b=1
f(x)=x**3; g(x)=x**2
n=25; i=0
load 're.plt'
plot [0:1] [0:1] f(x), g(x) lt 1


```



Obrázok 13

Teraz si ukážeme malú modifikáciu tohto príkladu. Stačí zmeniť

lt 0 na lt 1 (v súbore re.plt) a $n=25$ na $n=500$. (Pozri obrázok 14.)



Obrázok 14


Ako sme už videli v ukážkach, typ čiary možno voliť pomocou `linetype n`, kde n je vhodné číslo (najčastejšie používame $n \in \{0, 1, 2, 3, 4, 5, 6, 7, 8\}$). Používa sa aj skrátaná verzia `lt n`. Možnosti nastavenia zistíme v `gnuplot` príkazom `test`.

Pre danú funkciu $y = f(x)$ môžeme nakresliť jej integrál (ako funkciu hornej hranice), t. j. graf funkcie

$$g(x) = \int_a^x f(t) dt.$$

V ukážke použijeme funkciu $y = \cos x$. (Pozri obrázok 15.)

```
set border 31 lt 0
set xzeroaxis lt 4
f(x)=cos(x)
delta=0.02; a=0
integral(x,y)= x>y ? 0 : integral(x+delta,y)+delta*f(x)
g(x)= a<x ? integral(a,x) : -integral(x,a)
plot[0:2*pi] [-1:1] f(x), g(x)
```




Obrázok 15


Krivky dané parametricky

Nasledujúce príkazy nám poslúžia na nakreslenie asteroidy (pozri obrázok 16)

```
set parametric
set size square 0.5, 0.5
set zeroaxis
plot sin(t)**3,cos(t)**3
```



Obrázok 16



Obrázok 17

Lemniskátu nakreslíme pomocou príkazov (pozri obrázok 17)

```
set zeroaxis
set size square
set parametric
set samples 250
f(t)=t*(1+t**2)/(1+t**4)
g(t)=t*(1-t**2)/(1+t**4)
h(t)= t<1 ? g(t) : g(t-2)
k(t)= t<1 ? f(t) : -f(t-2)
set trange [-1:3]
plot k(t),h(t)
```

Klotoida

$$x(t) = \int_0^t \cos \frac{\pi s^2}{2} ds,$$


$$y(t) = \int_0^t \sin \frac{\pi s^2}{2} ds.$$

```
set samples 1000
set size square
set xzeroaxis
set yzeroaxis
f0(x)=cos(0.5*pi*x**2)
f1(x)=sin(0.5*pi*x**2)
integral0(x,y)=x>y?0:integral0(x+delta,y)+delta*f0(x)
```

```

integral1(x,y)=x>y?0:integral1(x+delta,y)+delta*f1(x)
g0(x)=a<x?integral0(a,x):-integral0(x,a)
g1(x)=a<x?integral1(a,x):-integral1(x,a)
delta=0.005; a=0
set param; set trange [0:4.9]
plot g0(t),g1(t),-g0(t),-g1(t) lt 1

```



Obrázok 18

Polárne súradnice


Nakreslíme si špirálu v polárnych súradniciach:

```

set polar
set function style impulses
plot t lt 2

```

Výsledok je na obrázku 19.




Obrázok 19

Krivka daná implicitne

Nakreslíme si kardioidu, ktorá je daná implicitnou rovnicou

$$(x^2 + y^2 - x)^2 = x^2 + y^2.$$

```
set isosamples 200
set xrange [-0.5:2.1]
set yrange [-1.3:1.3]
set contour base
set cntrparam levels discrete 0.0
set nosurface
set term table
set size square
set output 'krivka.dat'
plot (x**2+y**2-x)**2-x**2-y**2
set output; set term windows; set nokey; set nogrid; set noborder
set noxtics; set noytics; plot 'krivka.dat' with lines
```



Obrázok 20


Plochy

Dvojdielny hyperboloid

$$x^2 - y^2 - z^2 - 1 = 0$$

nakreslíme pomocou príkazov (pozri obrázok 21):

```
set parametric
set view 70, 165, 1, 1
set isosamples 70
set hidden3d
set xrange [ -3 : 3 ]
set yrange [ -3 : 3 ]
set zrange [ -3 : 3 ]
set arrow 1 from -3,0,0 to 3,0,0
set arrow 2 from 0,-3,0 to 0,3,0
set arrow 3 from 0,0,-3 to 0,0,3
f(u,v)=sqrt(1/v**2-1)*cos(u)
g(u,v)=sqrt(1/v**2-1)*sin(u)
plot [0:2*pi] [-1:1] 1/v, f(u,v), g(u,v) lt 0
```



Obrázok 21

Jednodielny hyperboloid


$$x^2 + y^2 - z^2 - 1 = 0$$

nakreslíme pomocou príkazov (pozri obrázok 22)

```

set parametric
set view 50, 135, 1, 1
set isosamples 90, 46
set hidden3d
set urange [0:2*pi]
set vrange [-1:1]
set xrange [-2:2]
set yrange [-2:2]
set zrange [-1.2:1.2]
set arrow 1 from -2.5,0,0 to 2.5,0,0
set arrow 2 from 0,-2.5,0 to 0,2.5,0
set arrow 3 from 0,0,-2.5 to 0,0,2.5
plot sqrt(1+v**2)*cos(u),sqrt(1+v**2)*sin(u),v lt -1

```




Obrázok 22

Valcová a guľová plocha v jednom obrázku (Vivianioh krivka, pozri obrázok 23):

```

set parametric
set hidden3d
set view 45, 130, 1, 1
set isosamples 90, 44
set urange [0:2*pi]
set vrange [0:pi/2]
set xrange [-1:1]
set yrange [-1:1]
set zrange [0:1]
gx(u,v)=cos(u)*cos(v)
gy(u,v)=sin(u)*cos(v)
gz(u,v)=sin(v)
vx(x)=0.5*(1+cos(x))
vy(x)=0.5*sin(x)
plot gx(u,v),gy(u,v),gz(u,v), vx(u),vy(u),v lt 1

```



Obrázok 23

Teraz si ukážeme, ako možno nakresliť graf funkcie $z = f(x, y)$, ktorá je definovaná na elementárnej oblasti

$$a \leq x \leq b$$

$$\varphi(x) \leq y \leq \psi(x)$$

Pre funkciu $z = \sin x + \cos y + 1$ definovanú na elementárnej oblasti

$$0 \leq x \leq \frac{3}{2}$$


$$\sqrt{1 - (1 - x)^2} \leq y \leq e^x$$

(nakreslíme si aj jej definičný obor) použijeme príkazy (pozri obrázok 24):

```

set parametric
set arrow 1 to 3,0,0
set arrow 2 to 0,3,0
set arrow 3 to 0,0,3
set isosamples 11,6
set view 45,320
dolna(x)=sqrt(1-(1-x)**2)
horna(x)=exp(x)
a=0; b=3./2.
f(x,y)=sin(x)+cos(y)+1
mix(x,y)=dolna(x)+(horna(x)-dolna(x))*y
set xrange [0:5]; set yrange [0:5]; set zrange [0:5]
set urange [a:b]; set vrange [0:1]
splot u,mix(u,v),0 lt 0, u,mix(u,v),f(u,mix(u,v)) lt 1

```



Obrázok 24

Hyperbolický paraboloid


$$x^2 - y^2 - z = 0$$

nakreslíme pomocou príkazov
(pozri obrázok 25)

```

set view 60, 30, 1, 1
set isosamples 75
set hidden3d
a=-1; b=1
set urange [a:b]

```



Obrázok 25

```

set vrange [0:1]
set xrange [-1.5:1.5]
set yrange [-1.5:1.5]
set zrange [-1.2:1.2]
dolna(x)=-sqrt(x*x+1)
horna(x)=sqrt(x*x+1)
f(x,y)=x*x-y*y
mix(x,y)=dolna(x)+(horna(x)-dolna(x))*y
splot u,mix(u,v),f(u,mix(u,v)) lt 1


```

Časť roviny $x - y + z - 1 = 0$ ležiacu vo štvrtom oktante nakreslíme pomocou príkazov (pozri obr. 26)

```

set arrow 1 to 1.3,0,0
set arrow 2 to 0,1,0
set arrow 3 to 0,0,1.2
set arrow 4 from 1,0,0 to 0,-1,0 nohead
set arrow 5 to 0,-1,0 nohead lt 0
set parametric; set view 60, 105
set urange [0:1]
set vrange [-1:0]
set xrange [-1.2:1.2]
set yrange [-1.2:1.2]
set zrange [0:1.2]
splot u,v,1-u+v lt 1

```



Obrázok 26

Nakreslíme si graf funkcie


$$f(x,y) = \max\left(\max(x,y), \varphi\left(\frac{x+y}{2}\right)\right),$$

kde

$$\varphi(x) = \begin{cases} 2x & x \leq \frac{1}{2} \\ 1 & \frac{1}{2} < x \leq 1 \\ x & 1 < x \end{cases}$$

Stačí použiť príkazy (pozri obrázok 27)

```
set border
set view 60,340
set isosamples 70
f1(x)= x<=0.5 ? 2*x : 1
f2(x)= x<=1 ? f1(x) : x
g(x,y)=max(max(x,y),f2((x+y)/2.))
max(x,y)= x<=y ? y : x
splot [0:2] [0:2] [0:2] g(x,y)
```




Obrázok 27

Krivky v priestore

Skrutkovicu nakreslíme pomocou príkazov (pozri obrázok 28)

```
set arrow 1 to 2,0,0
set arrow 2 to 0,1.2,0
set arrow 3 to 0,0,8
set view 60, 105
set isosamples 35
set param
set urange [0:2*pi]
set vrange [0:2*pi]
splot cos(u),sin(u),u lt 1
```



Obrázok 28

Na domácu úlohu si vyskúšajte nasledujúce príkazy:

```
set parametric; set samples 250; set isosamples 60, 30
f(t)=t*(1+t**2)/(1+t**4)
g(t)=t*(1-t**2)/(1+t**4)
h(t)=t<1?g(t):g(t-2)
k(t)=t<1?f(t):-f(t-2)
fx(u,v)=(u+1)/4.*cos(2*pi*v)
fy(u,v)=(u+1)/4.*sin(2*pi*v)
fz(u)=(u+1)/4.
set hidden3d; set view 45,0
splot [-1:3] [0:1] k(u),h(u),v, fx(u,v),fy(u,v),fz(u)
```

Odporúčam tiež navštíviť nasledujúcu adresu

<http://t16web.lanl.gov/Kawano/gnuplot/index-e.html>

Adresa autora:

Jozef Doboš, Katedra aplikovanej matematiky, SJF TU Košice

e-mail: Jozef.Dobos@tuke.sk, URL: <http://www.tuke.sk/dobos>

Jednota slovenských matematikov a fyzikov
Matematický ústav SAV
Fakulta prírodných vied Univerzity Konštantína Filozofa

OBZORY MATEMATIKY, FYZIKY A INFORMATIKY
4/2003(32)

Vydala Jednota slovenských matematikov a fyzikov v spolupráci
s vydavateľstvom PROTONIT s.r.o. s finančným príspevím
Ministerstva školstva Slovenskej republiky
Vedeckí redaktori: Beloslav Riečan, Daniel Klivanec
Výkonní redaktori: Peter Maličský, Aba Teleki
Technická a grafická úprava: Mária Benešová, Vladimír Kutnár
Všetky príspevky prešli odbornou recenziou

Podávanie novinových zásielok povolené
Západoslovenským riaditeľstvom pôšt Bratislava
č.j. 3001/2003-OLB zo dňa 1.10.2003

ISSN 1335-4981

OBSAH

| | |
|--|----|
| Jozef Doboš : gnuplot – grafický software (aj) pre matematikov | 1 |
| Anna Hapáková : Geometrické transformácie ako súčasť učebných osnov vo Veľkej Británii | 19 |
| Dušan Jedinák : Stručný prehľad objavov v matematike (Pohľad do dejín pre učiteľov školskej matematiky) | 24 |
| Daniel Kluvanec : Rok 2005: Medzinárodný rok fyziky WYP World Year of Physics | 31 |
| Arpád Kecskés, Aba Teleki : Hranica fyzikálneho poznania v súčasnosti..... | 36 |
| Peter Kluvánek : Paradoxy špeciálnej teórie relativity | 43 |
| Ivo Volf, Ľubomír Konrád : Úlohy riešené kvalifikovaným odhadom | 53 |
| Milan Turčáni : Internet a hypertext v prírodovednom vzdelávaní | 59 |
| Klement Hrkota : Dve poznámky ku kmitom pružinového kyvadla vo vyučovaní fyziky na strednej škole | 70 |

PRÍLOHA

FROM CONTENTS

| | |
|---|----|
| Jozef Doboš: gnuplot – Graphical Software (also) for Mathematicians | 1 |
| Anna Hapáková: Geometric Transformations as a Part of Teaching Schemes in the Great Britain | 19 |
| Dušan Jedinák: Compendium of Discoveries in Mathematics (Historical Preview for Teachers of School Mathematics)..... | 24 |
| Daniel Kluvanec: Year 2005: The World Year of Physics | 31 |
| Arpád Kecskés, Aba Teleki: Frontiers of Physics at the Present Time..... | 36 |
| Peter Kluvánek: Paradoxes of the Special Theory of Relativity | 43 |
| Ivo Volf, Ľubomír Konrád: Solutions of the Physics Problems with the Qualified Approximation | 53 |
| Milan Turčáni: Internet and Hypertext in the Science Education | 59 |
| Klement Hrkota: Two Notes on the Oscillation of the Spring Pendulum in the Teaching of Physics at Secondary School | 70 |